

## Excel Week 8 Lab – Amortization Table in Excel

In this week's lab activity, the goal is to create a schedule in Excel that breaks down every payment made on a 30-year mortgage into which portion goes to interest and which portion goes to principal. In a compound interest situation, interest begins to accrue at the end of the first month (at the time of the first mortgage payment). For this reason, a portion of each month's payment covers that interest.

What follows is a description from <https://www.wallstreetprep.com/knowledge/loan-amortization-schedule/>

### How to Calculate Amortization Schedule

The loan [amortization](#) schedule describes the allocation of interest payments and principal repayment across the maturity of the loan.

The borrower is required to fulfill payment obligations per the schedule laid out in the contractual agreement with the lender as part of the financing arrangement.

In particular, there are two forms of payment associated with loans: 1) the [interest expense](#) and 2) the [principal amortization](#).

- **Interest Expense** → The interest component reflects the cost of the borrowing, i.e. the credit risk associated with providing debt financing to the borrower is factored into the interest rate by the lender.
- **Principal Amortization** → The principal payments, on the other hand, represents the gradual repayment of the original principal over the maturity term.

Over the length of the borrowing term, the loan's book value gradually reduces in value until the outstanding balance reaches zero on the date of maturity.

If the balance does in fact reach zero, that means that the borrower met all of their debt obligations on time and did not default, i.e. did not miss an interest or principal payment.

What many people do not realize is that most of each month's payment goes to interest at first. Gradually the portion that goes to interest goes down and the portion that goes to principal goes up. However, it is a mistake people make thinking that their payments are helping them build equity in their home more than they really are.

The task this week is to create a schedule based on a 30-year fixed rate mortgage. (Note that this process would work for any fully amortizing loan). And the payment frequency is also the compounding periods per year.

**Mortgage amount: your choice of \$400,000 or \$350,000** ← type this into cell D4  
**Interest rate: your choice of either 5.0% or 5.5%** ← type this into cell D6  
**Borrowing term: 30 years** ← type this into cell D5  
**Payment frequency: 12** ← type this into cell H5

Also, please use the Excel template provided, which can be accessed through Office365, now Microsoft365. If desired it can be downloaded and you can work on it in Excel desktop application.

You will be introduced to two new functions in Excel:

“Interest Payment”:

“Principal Payment”:

IPMT(rate, per, nper, pv, [fv], [type])

and

PPMT(rate, per, nper, pv, [fv], [type])

IPMT(rate, per, nper, pv, [fv], [type])

The IPMT function syntax has the following arguments:

- **Rate** Required. The interest rate per period.
- **Per** Required. The period for which you want to find the interest and must be in the range 1 to nper.
- **Nper** Required. The total number of payment periods in an annuity.
- **Pv** Required. The present value, or the lump-sum amount that a series of future payments is worth right now.
- **Fv** Optional. The future value, or a cash balance you want to attain after the last payment is made. If fv is omitted, it is assumed to be 0 (the future value of a loan, for example, is 0).
- **Type** Optional. The number 0 or 1 and indicates when payments are due. If type is omitted, it is assumed to be 0.

Calculate the following:

1) In cell H4, calculate the monthly interest rate using cell references: **=D6/H5**

2) In cell H6, calculate the number of payments/periods: **=D5\*H5**

Now calculate the 1<sup>st</sup> month information across row 9. The loan amount in **D4** is the “present value” and the month in Cells **B9 through B368** is the current period or “**per**” argument.

Payment in cell **C9** **=PMT(\$H\$4,\$H\$6,\$D\$4)**

Interest in cell **D9** **=IPMT(\$H\$4,B9,\$H\$6,\$D\$4)**

Principal in cell **E9** **=PPMT(\$H\$4,B9,\$H\$6,\$D\$4)**

Balance in cell **F9** **=D4+E9** ← this changes in row 10 (do not fill this down)!

% Interest in cell **G9** **=D9/C9**

% Interest in cell **H9** **=E9/C9**

After the first month, the new balance will be calculated from the previous balance:

Balance in month #2 (in row 10, cell **F10**) type **=F9+E10**

**We want to keep the formatting** of the cells so before we do a “Fill/Down” we should enter formulas by hand into rows 10 and 11

Month #2 formulas should look like this:

Payment in cell **C10**                      =PMT(\$H\$4,\$H\$6,\$D\$4)  
Interest in cell **D10**                    =IPMT(\$H\$4,**B10**,\$H\$6,\$D\$4)  
Principal in cell **E10**                    =PPMT(\$H\$4,**B10**,\$H\$6,\$D\$4)  
Balance in cell **F10**                    =F9+E10  
% Interest in cell **G10**                    =D10/C10  
% Interest in cell **H10**                    =E10/C10

Month #3 formulas should look like this:

Payment in cell **C11**                      =PMT(\$H\$4,\$H\$6,\$D\$4)  
Interest in cell **D11**                    =IPMT(\$H\$4,**B11**,\$H\$6,\$D\$4)  
Principal in cell **E11**                    =PPMT(\$H\$4,**B11**,\$H\$6,\$D\$4)  
Balance in cell **F11**                    =F10+E11  
% Interest in cell **G11**                    =D11/C11  
% Interest in cell **H11**                    =E11/C11

Now SELECT cells C10 through H11 (you should have Months 2 and 3 selected) and FILL/DOWN all the way to ROW 368. If everything was done correctly, ROW 368 should **resemble this**:

8	Month	Payment	Interest	Principal	Balance	% Interest	% Principal
368	360	-\$2,147.29	-\$8.91	-\$2,138.38	\$0.00	0.4%	99.6%

(These numbers are for a mortgage amount of \$400,000 and interest rate 5% )

Finally, use the =SUM( ) function in Excel to total up the Payments, Interest, and Principal. Your spreadsheet should look like this after calculating the SUM of each column:

8	Month	Payment	Interest	Principal	Balance	% Interest	% Principal
367	359	-\$2,147.29	-\$17.78	-\$2,129.50	\$2,138.38	0.8%	99.2%
368	360	-\$2,147.29	-\$8.91	-\$2,138.38	\$0.00	0.4%	99.6%
369	Mortgage Repayment Components						
370	TOTALS:	-\$773,023.14	-\$373,023.14	-\$400,000.00			
371		Total Payment	Total Interest	Total Principal			

As part of this Lab Activity you will be asked some questions. Be prepared to answer them based on the schedule you have just created.